

Paper 060-2010

SMTP E-Mail Access Method: Hints, Tips, and Tricks

Chuck Hunley, SAS Institute Inc., Cary, NC

ABSTRACT

How do I specify multiple SMTP servers? What about authentication? Isn't this message important? How do I play a trick on a friend? Can I send a text message to a cell phone? This paper explores various hints, tips, and tricks using the FILENAME EMAIL SMTP (Simple Mail Transport Protocol) e-mail interface that answers all of these questions and more. Learn how to send an e-mail on behalf of your boss and send a message in HTML format.

INTRODUCTION

Sending e-mail with SAS is as simple as writing a DATA step. The SAS SMTP e-mail interface provides a way of sending a nicely formatted message that will capture the recipient's attention. This paper guides you from getting connected to an SMTP server to sending an eye-catching HTML e-mail message.

GETTING CONNECTED

Before you can send e-mail, you must have access to an SMTP server. Starting with SAS 9.2, you can specify multiple SMTP servers. When specifying a series of SMTP servers, enclose the list of servers in parentheses and each server name in single or double quotation marks.

Enabling the SMTP e-mail interface:

```
options emailsys = SMTP;
```

Specifying a single SMTP server:

```
options emailhost = smtp.domain.com;
```

Specifying multiple SMTP servers:

```
options emailhost = ("smtp1.domain.com" "smtp2.domain.com" "smtp3.domain.com");
```

If multiple SMTP servers are specified, then SAS attempts to connect to the servers in the order in which they are specified. If a connection to any of the specified servers is not established, then the e-mail is not delivered, and an error message is sent to the SAS log.

AUTHENTICATION

Some SMTP servers require that the user be authenticated before an e-mail is sent. If authentication is required, then the user ID and password that are specified must be for the SMTP server and not for the local host.

Enabling authentication:

```
options emailauthprotocol = LOGIN;
```

LOGIN is a protocol where the user ID and password are BASE64 encoded before sending the credentials to the SMTP server. Currently, only the LOGIN protocol is supported.

SPECIFYING THE PASSWORD

The user ID and password must be specified in plain text:

```
options emailpw="mypassword";
```

Note: Encoded password support is available in the third maintenance release for 9.2. For additional information about creating an encoded password, see PROC PWENCODE in SAS Help.

The SMTP e-mail interface authenticates the user in the following order:

1. FROM="userid@domain.com": If specified, *userid* is tried first. If unsuccessful, then *userid@domain.com* is tried.
2. EMAILID="userid@domain.com": If specified, *userid* is tried first. If unsuccessful, then *userid@domain.com* is tried.
3. If neither FROM= or EMAILID= is specified, or both fail to authenticate, then the SMTP e-mail interface looks up your user ID and domain from the operating system. The *userid* is tried first. If unsuccessful, then *userid@domain.com* is tried.

If authentication fails, then an error message is sent to the SAS log and the e-mail is not sent.

The following example looks up the user ID from the operating system:

```
options emailsys=smtp;
options emailauthprotocol=login;
options emailpw="mypassword";

filename myemail EMAIL
to="first.last@domain.com"
subject="Coder's Corner";

data _null_;
  file myemail;
  put "Please attended Coder's Corner";
run;

filename myemail clear;
```

To disable authentication:

```
options emailauthprotocol = NONE;
```

PRIORITY


The priority or importance of an e-mail message is specified in the FILENAME or FILE statement, or with the PUT statement !EM_IMPORTANCE! directive. A message can have a priority of LOW, NORMAL, or HIGH. E-mail clients, such as Microsoft Outlook, display an indicator of the message priority.

Setting a HIGH priority:

```
filename myemail EMAIL
to="first.last@domain.com"
from="SAS Global Forum 2010 <sgf2010@sas.com>"
importance="HIGH"
subject="SAS Global Forum 2010";


data _null_;
  file myemail;
  put "Welcome to SAS Global Forum 2010";
  put " ";
  put "Please attended Coder's Corner";
run;

filename myemail clear;
```



Setting a LOW priority:

```
filename myemail EMAIL
to="first.last@domain.com"
from="Joe Manager <Joe.Manager@sas.com>"
importance="LOW"
subject="Status Report";
```



Setting a HIGH priority with the PUT statement !EM_IMPORTANCE! directive:

```
filename myemail EMAIL;

data _null_;
  file myemail;
  put "!EM_TO!" 'first.last@domain.com';
  put "!EM_SUBJECT! Status Report";
  put "!EM_IMPORTANCE! High";
  put "Please send your weekly status report";
  put " ";
  put "Thanks!";
  put "Joe";
run;

filename myemail clear;
```

TIME FOR A TRICK!

Have you ever wanted to fool a friend or colleague? You can specify the FROM e-mail address so that the message appears to have come from another person. This common spammer's technique can be used to send a valid message on behalf of someone else or to create a NOREPLY e-mail address for the message.

SPECIFYING THE SENDER

The sender's name can be specified in addition to the e-mail address. The format is "*Firstname Lastname* <userid@domain.com>". If the recipient's e-mail client supports this format, then the "*Firstname Lastname*" is

displayed instead of the e-mail address. The SENDER does not need be the same, though making it your e-mail address helps in cases where a message is undeliverable. Undeliverable messages get returned to the sender.

```
filename myemail EMAIL
to="userid@domain.com"
from="Mickey Mouse <first.last@domain.com>"
sender="Mickey Mouse <first.last@domain.com>"
subject="Seattle?";

data _null_;
  file myemail;
  put "Don't go to Seattle, come to Disney World!";
run;

filename myemail clear;
```

HTML IS NOT JUST FOR WEB PAGES

Plain text e-mail can deliver your message to the recipient, but can be ugly and boring. By sending a message in HTML format, you can add some creativity to your message by adding links, images, color, and formatted text to capture your recipient's attention. The SMTP e-mail interface method does not convert plain text into HTML. So, the HTML has to be created before sending the message. Once the HTML is created, the e-mail content type can be set to TEXT/HTML. Setting the content type informs the recipient's e-mail client how to display the message.


```
options emailsys=smtp emailhost=smtp.domain.com;

filename myemail email
to="chuck.hunley@sas.com"
from="SAS Global Forum <sgf2010@sas.com>"
sender="SAS Global Forum <sgf2010@sas.com>"
subject="SAS Global Forum"
content_type="text/html";

data _null_;
  file myemail;
  put '<html>';
  put '<head>';
  put '<meta content="text/html; charset=ISO-8859-1" http-equiv="content-type">';
  put '<title>SAS Global Forum</title>';
  put '</head>';
  put '<body>';
  put '<b>';
  put '<span style="font-size: 12pt;';
  put 'font-family: &quot;Times New Roman&quot;, &quot;serif&quot;;">';
  put '<br>';
  put 'Welcome to SAS® Global Forum!';
  put '<br><br>';
  put '';
  put '</span>';
  put '</b>';
  put '</body>';
  put '</html>';
run;

filename myemail clear;
```

The resulting e-mail should be displayed as follows in HTML format in supporting e-mail clients:



ATTACHMENTS

Files can be attached to an e-mail message. This is done with the ATTACH=FILENAME option. The syntax is:

```
ATTACH = "filename.ext" | ATTACH = ("filename.ext" <attachment-options>)
```

ATTACHMENT OPTIONS

CONTENT_TYPE | CT | TYPE="content/type"

Specifies the content type of the attached file. If the content type is not specified, SAS tries to determine the correct content type based on the filename extension. When trying to determine the correct content type, SAS first looks for the filename extension in the SAS Registry under SAS_REGISTRY\COREFILETYPES. If it is not found in the SAS Registry and you are on a Microsoft Windows operating system, then the Windows Registry is queried. If no content type can be determined, then TEXT/PLAIN is used as the default. Additional content types can be added to the SAS Registry.

EXTENSION | EXT = "extension"

Specifies a different filename extension. The filename extension is used by some e-mail clients to determine how to display the attachment. For example, the following changes the original filename from home.html to home.htm:

```
ATTACH=("home.html" EXT="htm")
```

LRECL=record length

Specifies the maximum record length in the attachment. The default LRECL for attachments is 256 characters. If the attachment has longer lines, specify a larger LRECL value.

```
ATTACH=("home.html" LRECL=32727)
```

NAME="filename"

Specifies a different filename to be used for the specified attachment. For example, the following changes the attachment name from home.html to index.html:

```
ATTACH= ("home.html" NAME="index")
```

ENCODING="encoding-value"

Specifies the text encoding of the attachment that is read into SAS. Use this option when the encoding of the attachment on disk is different from your SAS session encoding. (See the *SAS National Language Support (NLS): Reference Guide*.)

OUTENCODING="encoding-value"

Specifies the resulting text encoding for the attachment to be sent. Use this option when your SAS session encoding is different from the attachment encoding that you want. (See the *SAS National Language Support (NLS): Reference Guide*.) For example, if a text attachment in z/OS is in EBCDIC, and the recipient needs to receive it as an ASCII-based encoding, the syntax looks like the following:

```
ATTACH= ("report" EXT="txt" ENCODING="ebcdic1047" OUTENCODING="latin1")
```

RECFM=record format

Specifies the record format of the attachment to be read into SAS. The RECFM values for attachments are the same as the FILENAME RECFM value for DISK files.

Multiple attachments can be specified. A filename enclosed in quotation marks represents the boundary for the start of the next attachment. For example, here is the syntax for specifying multiple attachments:

```
ATTACH= ("home.html" NAME="letter" EXT="htm" LRECL=32767 "report.xls"
LRECL=1024)
```

TEXTING

Sending text messages via e-mail can be problematic. It can be done, but it requires some investigation as each cell phone service provider has different criteria. Most cell phone companies have special addresses (domains) to which an e-mail can be sent. The typical e-mail address format is 10digitphonenumber@domain.com. The message is relayed to the recipient's cell phone. Additionally, each provider has limitations on message length and FROM length. Some have limited means for preventing spam messages and require that both the FROM and SENDER e-mail addresses match.

```
filename myemail EMAIL
to="5553219876@vtext.com"
from="first.last@domain.com"
sender="first.last@domain.com"
subject="SAS Global Forum 2010";
```

```
data _null_;
file myemail;
put "SAS Global Forum 2010";
put "Seattle, Washington";
put "April 11-14";
run;
```

```
filename myemail clear;
```

WHEN ALL ELSE FAILS

If you have problems sending e-mail, then the FILENAME or FILE statement DEBUG option can help. When the DEBUG option is specified, communication between SAS and the SMTP server is sent to the SAS log:

```
filename myemail EMAIL DEBUG;
```

Lines prefixed with **s**: represent those sent, and lines prefixed with **R**: represent those received.

```
R: 220 smtp.domain.com [ESMTP Server] service ready;_ ; 01/22/10 15:55:42
S: HELO mymachine.domain.com.com
```

```

R: 250 smtp.domain.com
S: MAIL FROM:<userid@domain.com>
R: 250 <userid@domain.com> Sender OK
S: RCPT TO:<First.Last@domain.com>
R: 250 <First.Last@domain.com> Recipient OK
S: DATA
R: 354 Start mail input; end with <CRLF>.<CRLF>
S: Date: 22 Jan 2010 15:55:42 -0500
S: Subject: SAS Global Forum Photo
S: From: userid@domain.com
S: To:
S: First.Last@domain.com
S:
S: X-Mailer: 9.02.02M2D092309
S: MIME-Version: 1.0
S: Content-Type: multipart/mixed; boundary="NEXTPART"
S:
S: --NEXTPART
S: Content-Type: text/plain; charset=iso-8859-1
S: Content-Transfer-Encoding: 8bit
S:
S: Hello!
S:
S:
S:
S: Here is a photo from SAS Global Forum 2010!
S:
S:
S: --NEXTPART
S: Content-Type: image/gif;
S:     name="photo.gif"
S: Content-Transfer-Encoding: base64
S: Content-Disposition: attachment;
S:     filename="photo.gif"
S:
S: R0lGODlhRgIJAKEAAG9vb+fn5wAA/wAAACH5BAEAAAIALAAAAABGAgkAAAK5lI+pK+DA2ou0
S: 2ouz3rz7D4biSJbmiabqyiaT9bbyTNf2jef6zvdqbAD6hsSi8YhMKpe3RwASdDyZ1Kr1is1q
S: txRnYOqdcsfksvmMNkOj3y9E2l5Dl+m6/Y7P6yMOiRzwBPgmKBjlt4eYqLjIaPLwCBkp+SIV
S: Qzq3mam5ydnP+QkaKjPkwmp6ipqqusra6voKGwvXRltre/sVlYcQg+v7CxwsPExcbHyMnKy8
S: zNzs/AwdLTlNXW3dVgAAOw==
S:
S: --NEXTPART--
S: .
R: 250 OK: <5a230a64001442d4@smtp.domain.com >
S: QUIT
R: 221 [ESMTP Server] service closing transmission channel

```

MESSAGE SENT!

Your e-mail messages don't have to be just simple! The SAS SMTP e-mail interface can be a powerful and useful tool in your communication arsenal. So, make your messages of the highest importance, let them stand out, and communicate globally!

REFERENCES

"SAS OnlineDoc 9.2 PDF Files." SAS OnlineDoc 9.2: PDF Files, Second Edition. 2010. CD-ROM. SAS Institute Inc., Cary, NC.

RECOMMENDED READING

"SAS OnlineDoc 9.2 PDF Files." SAS OnlineDoc 9.2: PDF Files, Second Edition. 2010. CD-ROM. SAS Institute Inc., Cary, NC.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Chuck Hunley
SAS Campus Drive
SAS Institute Inc.
E-mail: chuck.hunley@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.